


# Wurzelspirale

Prof. Dr. Dörte Haftendorn, Juni 09 Update 19.06.09

[www.mathematik-verstehen.de](http://www.mathematik-verstehen.de) <http://haftendorn.uni-lueneburg.de>


#####

```
sp:=plot::Polar([sqrt(t),t],t=0..20*PI,  
Mesh=500):plot(sp)
```




Zugehörige Kartesische Darstellung


```
plotfunc2d(sqrt(t), sqrt(2*k*PI))$  
k=1..10,sqrt(2*9*PI),t=0..20*PI, LegendVisible=FALSE)
```



Hier sieht man, dass sie nach außen zu immer enger wird.  
Darstellung der Abstände

```
plotfunc2d(sqrt(t+2*PI)-sqrt(t),t=0..300, Axes=Origin,  
ViewingBoxYRange=0..3)
```





Die Breite der Spiralenringe geht gegen Null. Die Breite der gesamten Wurzelspirale ist aber nicht beschränkt.

#####


```
r:=t->sqrt(t);
```

$$t \rightarrow \sqrt{t}$$

```
steig:=t->(r'(t)*sin(t)+r(t)*cos(t))/(r'(t)*cos(t)-r(t)*sin(t))
```

$$t \rightarrow \frac{r'(t) \cdot \sin(t) + r(t) \cdot \cos(t)}{r'(t) \cdot \cos(t) - r(t) \cdot \sin(t)}$$

```
plotfunc2d(steig(t), t=0..10)
```



Die einzige Wendestelle ist als Extremstelle gut zu sehen.  
Nenner und Zähler extra:

```
denom(steig(t)), numer(steig(t))
```

```
denom(steig(t)), numer(steig(t))
-2*sqrt(t)*(cos(t)-2*t*sin(t)), -2*sqrt(t)*(sin(t)+2*t*cos(t))
```

Polarwinkel, bei denen waagerechte Tangenten vorliegen.

```
numeric::solve(numer(steig(t))=0,t=0);
numeric::solve(numer(steig(t))=0,t=2);
numeric::solve(numer(steig(t))=0,t=5);
{0.0}
{1.836597203}
{4.815842318}
```

Polarwinkel, bei denen senkrechte Tangenten vorliegen.

```
numeric::solve(denom(steig(t))=0,t=1);
numeric::solve(denom(steig(t))=0,t=3);
numeric::solve(denom(steig(t))=0,t=6);
{0.6532711871}
{3.292310021}
{6.361620392}
```

Steigungen in den Nullstellen

```
simplify(steig(k*PI)) $ k=1..10;
2*pi, 4*pi, 6*pi, 8*pi, 10*pi, 12*pi, 14*pi, 16*pi, 18*pi, 20*pi
```

Steigungen in den Durchgängen durch die y-Achse

```
simplify(steig((2*k+1)/2*PI)) $ k=1..10;
-1/(3*pi), -1/(5*pi), -1/(7*pi), -1/(9*pi), -1/(11*pi), -1/(13*pi), -1/(15*pi), -1/(17*pi), -1/(19*pi), -1/(21*pi)
```

Rechts und links werden die Nulldurchgänge immer steiler, oben und unten immer flacher.

#####

Betrachtung der Fläche

```
item:=x->1/2*int(t, t=0..2*k*PI): item(x)
pi^2*k^2
```

Bei der polaren Flächenbestimmung wird bei jeder Runde das Innere ganz überstrichen. Daher ist bei k Runden nur der letzte Ring einmal in der mit dem Integral berechneten Fläche, der vorige doppelt, davor dreifach und so weiter.

```
t1:=0:item(2*k*PI) $ k=1..4
pi^2, 4*pi^2, 9*pi^2, 16*pi^2
```

```


sc1:=plot::Scene2d(plot::Polar([sqrt(t),t],t=0..2*PI,
Mesh=500)):
sc2:=plot::Scene2d(plot::Polar([sqrt(t),t],t=0..4*PI,
Mesh=500)):
sc3:=plot::Scene2d(plot::Polar([sqrt(t),t],t=4*PI..8*PI,
Mesh=500)):
sc4:=plot::Scene2d(plot::Polar([sqrt(t),t],t=6*PI..10*PI
, Mesh=500)):

```

```

plot(sc1,sc2,sc3,sc4)

```



Die 1. Fläche bis 3. Flächen der Ringe sind

```

PI^2, 4*PI^2-2*PI^2, 9*PI^2-2*(4*PI^2-2*PI^2)-3*PI^2
π2, 2·π2, 2·π2

```

Die 4. Fläche

```

PI^2*(16-2*2-3*2-4*1)
2·π2

```

Die k-te Fläche

```

PI^2*(k^2-2*sum(k,k=2..(k-1))-k*1);
expand(%)
-π2·(k+k·(k-1)-k2-2)
2·π2

```

Erstaunliches Ergebnis: alle die Spiralenringe haben den Flächeninhalt

$$2 \cdot \pi^2$$

#####

Betrachtung von Kreisringen, die auch durch die rechten

Betrachtung von Kreisringen, die auch durch die rechten Achsenschnittpunkte gehen.

```
PI*2*(k+1)*PI-PI*2*k*PI;
```

```
expand(%)
```

```
2·π2·(k+1) - 2·π2·k
```

```
2·π2
```

Auch diese Ringe haben in jeder Größe den Flächeninhalt  $2 \cdot \pi^2$ .

```
kr1:=plot::Circle2d(sqrt(2*k*PI)|k=1,[0,0],
LineColor=[1,0,0]):
```

```
kr2:=plot::Circle2d(sqrt(2*k*PI)|k=2,[0,0],
LineColor=[1,0,0]):
```


```
sc1:=plot::Scene2d(kr1,kr2,plot::Polar([sqrt(t),t],t=2*PI..6*PI,
Mesh=500)):
```

```
kr3:=plot::Circle2d(sqrt(2*k*PI)|k=2,[0,0],
LineColor=[1,0,0]):
```

```
kr4:=plot::Circle2d(sqrt(2*k*PI)|k=3,[0,0],
LineColor=[1,0,0]):
```

```
sc2:=plot::Scene2d(kr3,kr4,plot::Polar([sqrt(t),t],t=4*PI..8*PI,
Mesh=500)):
```

```
plot(sc1,sc2)
```



```
kreise:=(plot::Circle2d(sqrt(2*(k-1)*PI),
LineColor=[1,0,0]) $ k=1..11):
```

```
sc1:=plot::Scene2d(kreise):
```

```
sc2:=plot::Scene2d(kreise,sp):
```

```
plot(sc1,sc2)
```

